


AN 246-2



Measuring Phase Noise With
The HP 3585A Spectrum Analyzer


Introduction

There are a variety of systems in common usage today whose performance is affected directly by the frequency stability of their internal signal source. Examples of these are Doppler radars, data communication links, and multi-channel receivers. Complete characterization of signal source frequency stability includes short-term phase shift caused by both coherent signals and random phenomena com-

monly called phase noise. Detecting and measuring phase noise is the subject of this application note. This topic will be covered beginning with theory and ending with a practical BASIC program listing for automatic measurements using the HP 3585A Spectrum Analyzer and the HP 9845T Desktop Computer. An abbreviated program and listing for the HP 85A Personal Computer is also included.



HP 9845T



HP 3585A



HP 85A

Contents

| | Page |
|--|-------------|
| Chapter 1 – Gaining an Intuitive Understanding | 4 |
| Chapter 2 – Effects of Phase Noise on Real Systems | 5 |
| Chapter 3 – Relating Phase Noise to Frequency Stability | 6 |
| Chapter 4 – Practical Methods of Measuring Phase Noise | 8 |
| Appendix A – Software Description | 11 |
| Appendix B – Accounting for Analyzer Characteristics | 15 |
| References | 16 |

Chapter 1

Gaining an Intuitive Understanding

A brief review of the components of frequency stability is helpful in developing a working understanding of phase noise. There are three fundamental elements involved.


The first, long-term stability, is frequency changes which occur over long periods of time. This is usually expressed in terms of parts per million of frequency change per hour, day, month, or year. It represents a reasonably predictable phenomenon due to the aging process of the material used in the frequency determining element. In quartz oscillators, for example, it is the crystal element which determines the long-term frequency stability.

The second, environmentally induced frequency shifts, is comprised of effects related to temperature, pressure, and even gravity. Since these effects are a stimulus-response phenomenon, they are relatively easy to isolate and measure quantitatively. With proper design, most of these effects are secondary in importance. Using the quartz oscillator again as an example, it is a common practice to place the crystal in a temperature controlled oven to reduce the amount of temperature related frequency shifts.

The third component, short-term frequency fluctuations, contains all elements causing frequency or phase changes about the nominal frequency of less than a few seconds duration. Determining the composition of these fluctuations is the key to understanding short-term frequency stability. Generally there are two categories of frequency

fluctuations. The first, deterministic, includes discrete signals which can be easily related to known phenomena such as power line frequency, vibration frequencies, or AC magnetic fields. These components show up as discrete modulation sidebands on the fundamental signal. The second, random fluctuations, are best described in terms of a statistical distribution. This distribution, commonly called phase noise, is usually measured and presented as a spectral density plot of the modulation sidebands in the frequency domain. The term, spectral density, means that the frequency domain energy distribution is a continuous function. Therefore, the amplitude of the sidebands must be expressed in terms of energy within a specified bandwidth, normally a one hertz bandwidth. Figure 1a shows a typical RF sideband spectrum with both deterministic sidebands and noise sidebands.

It is important to note here that the modulation sidebands on the fundamental are directly proportional to an equivalent phase modulating source only if two conditions are met. First, sideband energy due to amplitude modulation must be much less than due to phase modulation and second, the total phase deviations must be less than one radian. The second condition is another way of saying that the higher order sidebands which result from frequency modulation must be small in order for the sidebands to directly represent the equivalent modulating source shown in Figure 1b. Fortunately, this is the case for most stable frequency sources.



Chapter 2

The Effects of Phase Noise on Real Systems

With a basic mental image of phase noise in hand, it is appropriate to look at the effects on practical systems.

Radar systems have been greatly advanced since their early years and depend on highly sophisticated techniques for increased range and resolution. Doppler radars, for example, utilize a narrow bandwidth receiver to detect the shifted frequency return of a moving target relative to the return from the ground. The total power of the ground return, called ground clutter, far exceeds the power of the target return, thus, the need for the narrow bandwidth receiver tuned to the target return frequency. Phase noise on either the transmitter oscillator or the receiver local oscillator can limit range resolution and sensitivity.

In terms of resolution, the receiver bandwidth is limited to that which will pass the majority of the return frequency energy. Phase noise spreads the target energy, thus requiring a wider IF bandwidth. Sensitivity is limited by ground clutter energy which appears in the receiver IF bandwidth. Excessive phase noise effectively smears the clutter energy into the receiver IF bandwidth. Requirements for phase noise on radar systems will typically be -110dB/Hz referred to the transmitter level and the local oscillator level from a few hertz up to several hundred kilohertz away from the carrier. See Figure 2.

Phase-modulated data systems are also sensitive to phase noise which is phase detected along with the desired modulation signal. The effect is to degrade the bit error rate performance. The maximum allowable phase noise is usually specified in terms of the total equivalent rms degrees of phase noise modulation within the data channel bandwidth. For values of phase noise less than 5 degrees rms, the effect on bit error is directly additive to the receiver thermal noise. Typical system specifications call for equivalent phase noise modulation of less than 2 degrees rms from 20kHz to 80MHz from the carrier for a 33.5GHz carrier.

In general, multi-channel communications receivers have a problem directly related to phase noise on their local oscillators. LO sidebands appear on the received signal in the IF at the same ratio as they exist on the LO. For example, phase noise sidebands 100dB/Hz down from the LO will appear 100dB/Hz down on the received signal. This presents no problem for a single channel receiver since that ratio is more than adequate for intelligibility. However, in a multi-channel receiver, the sensitivity with a strong signal in an adjacent channel is set by the level of the phase noise sidebands at an offset equal to the channel spacing. Figure 3 shows this relationship pictorially.




Figure 2. Doppler Radar




Figure 3. Adjacent Channel Sensitivity

Chapter 3

Relating Phase Noise to Frequency Stability

The mathematical relationship of phase noise to frequency stability has been rigorously treated in a series of technical notes by the U.S. National Bureau of Standards. This chapter presents a brief summary of their conclusions.

An ideal sinewave source can be described by:

$$V(t) = V_o \sin 2\pi\nu_o t$$

where V_o = nominal amplitude

ν_o = nominal frequency

In the real world, of course, there are fluctuations in both amplitude and frequency which can be represented by the following additional terms.

$$V(t) = [V_o + \epsilon(t)] \sin [2\pi\nu_o t + \phi(t)]$$

where $\epsilon(t)$ = amplitude fluctuations

$\phi(t)$ = phase fluctuations

Long-term amplitude and phase fluctuations are best described in terms of time but fluctuations with periods of less than a few seconds are easier to understand when translated by Fourier expansion into the frequency domain. Figure 4 is an example of the RF spectral density of a sinewave source showing phase noise and discrete AM and PM sideband components. Note the unequal levels of the symmetrically spaced discrete sideband signals. This is a good indication that both AM and PM are taking place. The AM components can be eliminated and the PM sidebands demodulated by phase detecting the fundamental signal. The resulting signal when viewed in the frequency domain is the spectral density of the equivalent modulating noise source.

$$S_{\phi}(f) = \frac{S_{V_{rms}}(f)}{K^2} \left[\frac{\text{rad}^2}{\text{Hz}} \right]$$

where $S_{V_{rms}}(f)$ = the power spectral density of the voltage fluctuations out of the phase detector

K = phase detector constant (volts/radian)

This spectral density is particularly useful for analysis of phase noise effects on systems which have phase sensitive circuits such as digital FM communication links.




Figure 4. AM and PM Sidebands

The spectral density of frequency fluctuations is also an important quantity and can be easily derived from the phase spectral density. Since frequency is the time rate of change of phase, it follows that for $\nu(t)$ being the time function of frequency:

$$\begin{aligned} 2\pi\nu(t) &= \frac{d(\phi(t) + 2\pi\nu_0 t)}{dt} \\ &= \frac{d\phi}{dt} + 2\pi\nu_0 \\ \text{and } \nu(t) &= \frac{1}{2\pi} \frac{d\phi}{dt} + \nu_0 \end{aligned}$$

From transform theory the Fourier expansion of $\nu(t)$ is:

$$\nu(s) = \frac{s}{2\pi} \phi(s)$$

and the spectral density is:

$$S_\nu(f) = \frac{(2\pi f)^2}{(2\pi)^2} S_\phi(f) = f^2 S_\phi(f) \quad \left[\frac{\text{Hz}^2}{\text{Hz}} \right]$$

Caution must be taken when using the phase noise of sources at different frequencies. $S_\nu(f)$ is the spectral density of absolute frequency fluctuations. The measured spectral density of a 10MHz source would represent a much greater percentage frequency fluctuation than the same spectral density if measured at 100MHz. The answer to this problem is the spectral density recommended by the U.S. National Bureau of Standards as the primary definition of frequency stability. The spectral density of fractional frequency fluctuations, $S_Y(f)$, is related to frequency fluctuations and phase fluctuations by:

$$S_Y(f) = \frac{1}{\nu_0^2} S_\nu(f) = \frac{f^2}{\nu_0^2} S_\phi(f) \quad \left[\frac{1}{\text{Hz}} \right]$$

In many cases, however, it is not the spectral density of the equivalent modulating source that is of interest but rather the actual sideband power of phase fluctuations with respect to the carrier level. As an expression, this is:

$$\mathcal{L}(f) = \frac{\text{Power density (one phase modulation sideband)}}{\text{Carrier Power}}$$

This spectral density, Script $\mathcal{L}(f)$, is defined as the ratio of the power per hertz of bandwidth at a frequency f from the carrier in one phase noise sideband to the carrier power. For sideband levels within the dynamic range of wave or spectrum analyzers,

this quantity can be measured directly at the RF frequency with one assumption. The AM components must be small compared to the PM components. Fortunately, this is the case for most sources of frequency standard quality.

For sources with very low sidebands, it is necessary to use phase detection or frequency discrimination to effectively eliminate the carrier in order to gain measurement range. Here another assumption must be made. The phase detected spectral density will be equivalent to the actual sidebands only if the peak phase fluctuations are much less than one radian. In terms of frequency modulation theory, this is equivalent to saying that the higher order modulation components are insignificant compared to the fundamental modulating frequency. For nearly all high quality sources, this is a good assumption. Given this assumption, $\mathcal{L}(f)$ can be related to $S_\phi(f)$. From small angle modulation theory:

$$\begin{aligned} \mathcal{L}(f) &= \left[\frac{\phi_{\text{peak}}}{2} \right]^2 \quad (1\text{Hz BW}) \\ &= \left[\frac{1.4\phi_{\text{rms}}}{2} \right]^2 \quad (1\text{Hz BW}) \\ &= \frac{S_\phi(f)}{2} \end{aligned}$$

Another common expression, signal to phase noise ratio, is the rms value of the phase noise sidebands in a specified bandwidth about the carrier with respect to the carrier power. A commonly used bandwidth is 30kHz centered on the carrier excluding $\pm 1\text{Hz}$ around the carrier. This expression is convenient because it yields a single number but it is not as informative as a spectral density.

Frequency stability is also defined in the time domain with a sample variance known as the Allan variance. The expression is usually simplified to:

$$\sigma_Y^2(\tau) = \frac{(Y_{k+1} - Y_k)^2}{2}$$

$$\text{where } Y_k = \frac{\phi(t_k + \tau) - \phi(t_k)}{2\pi\nu_0\tau}$$

τ = repetition interval

This measurement is usually made with a counter system and is particularly useful because a transformation to the frequency domain yields data closer to the carrier than possible with most currently available frequency analyzers. A description of the transformation can be found in Reference 3.

Chapter 4

Practical Methods of Measuring Phase Noise

Measurement of phase noise sidebands would be simple if spectrum analyzers had dynamic ranges of 160dB and 1Hz bandwidths useable in the GHz region. All measurements could then be made at the primary frequency of the source. However, equipment with this performance does not exist and so alternate techniques must be used. This chapter describes the RF spectrum measurement and two alternatives: frequency discrimination and quadrature phase detection.

Direct RF Spectrum Measurement




As mentioned before, the sidebands of a signal may represent both AM and PM. Asymmetry in the sidebands is an indication that both AM and PM are present. However, in many cases, due to the manner in which the signal has been processed, PM sidebands are dominant. For example, if a reasonably clean synthesized signal is multiplied up to be used as a high frequency reference, the phase noise sidebands are multiplied by the same factor as the frequency while the AM sidebands are not changed or are limited. In this case, direct RF spectrum measurements at the multiplied frequency are a good approximation of the phase noise sidebands. The sidebands, when corrected and normalized to the carrier powers, represent the $S_{\nu}(f)$ spectral density described in Chapter 3.

One way to achieve better resolution is to translate the signal down in frequency to the range of an analyzer with the desired IF bandwidth. Figure 5 shows a typical setup using a doubly balanced mixer and a low pass filter. One of the advantages of this technique is that AM sidebands on the measured signal will be stripped off if it is to be used as the high level signal at the mixer. Two potential problems must be considered as well. First, the difference frequency will contain sidebands which are

folded up from below zero frequency. Whether or not the sidebands are significant depends on the nature of the particular source being measured. The second problem is that phase noise sidebands from the reference frequency at the mixer will also be translated down. This problem is avoided by using a source with better phase noise specifications than the one being tested.

Frequency Discrimination

The only way to solve the problem of measuring sidebands which are beyond the dynamic range of the analyzer is to eliminate the carrier frequency. One way to do that is with a frequency discriminator as shown in Figure 6. It is necessary to check the linearity of the discriminator over the frequency range of interest to insure that the calibration factor is constant. For microwave frequencies, the cavity discriminator is particularly useful for this type of measurement. Additional information on this type of measurement is contained in Reference number 1.



Quadrature Phase Detection

Perhaps the most versatile setup (Figure 7) is a doubly balanced mixer with the unknown source and reference source set in phase quadrature (90°) at the input. At quadrature, the difference frequency is zero hertz and the average voltage output is zero volts. For phase fluctuations $\ll 1$ radian the voltage fluctuations at the mixer output are related to the phase fluctuations by the equation:

$$\phi = \frac{v}{K}$$

where K = calibration factor in volts/radian

The system is easily calibrated by offsetting one of the sources and observing the resultant beat signal on an oscilloscope. The slope at the zero crossing in volts/radian is K and for sinusoidal beat signals (harmonics < 40 dB) is equal to the peak voltage of the signal. The beat signal as viewed on an analyzer is the rms value and so is 3dB less than the peak. In terms of the ratio of the sideband voltage to the beat signal voltage:

$$10 \log_{10} (S\phi(f)) = V_S - (V_B + 3\text{dB})$$

$$10 \log_{10} (\mathcal{L}(f)) = V_S - (V_B + 3\text{dB}) - 3\text{dB} \quad \phi(f) \ll 1$$



where V_S = sideband voltage in dBV corrected for bandwidth and analyzer characteristics

V_B = beat signal rms level in dBV

The underlying assumption so far is that the reference source has much lower phase noise than the unknown source. For state of the art sources, it is possible to compare two "identical" sources and assume that the phase noise of either one is 3dB less than the measured values. Measurement of various combinations of pairs of "identical" sources will test this assumption.

Often the long-term stability of sources is not sufficient for a quadrature phase relationship to be held during the measurement period. If this is the case, one of the sources must be adjusted periodically. A phase lock loop as shown in Figure 8 may be used if one or both of the sources have a voltage control for small frequency adjustments. In order to retain a constant relationship between phase and voltage fluctuations, the low frequency cutoff of the phase lock loop must be below the lowest frequency to be analyzed. If the breakpoint is moved out by adding gain in the loop, the voltage fluctuations at frequencies below the breakpoint will represent frequency fluctuations. Calibration using the phase lock setup is done by disconnecting the feedback voltage and observing the beat signal as before.

The following example of phase noise analysis of a 10MHz synthesized source shows how the HP 3585A Spectrum Analyzer and a controller such as the HP 9845T Desktop Computer can be used to make automated measurements. The basic setup is the phase locked heterodyne circuit shown in Figure 9.



Although the HP 3585A can be used in the conventional spectrum analysis mode for phase noise measurements, there are several limitations. The most obvious is that the linear frequency sweep does not provide sufficient frequency resolution in a single sweep over the several decades of frequency coverage usually desired. In addition, the video filtering which is required to reduce the variance of the input voltage fluctuations results in very long sweep times. These limitations can be overcome by programming the HP 3585A to make spot measurements at logarithmically spaced frequencies using only enough points for the resolution desired. When making noise measurements with any general purpose spectrum analyzer, error correction and bandwidth normalization must be applied to obtain an

accurate measurement; however, with the 3585A it is not necessary when the noise level function is used with the marker display. Not only is the correction and normalization performed automatically but the signal is measured 100 times and digitally averaged to reduce the variance of the final reading. All these features greatly simplify writing a program for automated phase noise measurements.

Using a versatile controller such as the HP 9845T Desktop Computer with BASIC language provides the user with more than just automated measurements over the HP-IB interface bus. The entire phase noise measurement procedure can be integrated into a program which guides the operator from measurement calibration to hard copy output of final results. The software listing included here is an example of a general purpose program which can be used without changes for most phase noise measurements. It can be easily streamlined by deleting the portions which do not apply to a particular measurement application.

There are four parts (subprograms) to the program which must be run in sequence for complete measurement results. However, it is not necessary to go back to the beginning to repeat any of the four subprograms. The first part, called CALIBRATE, guides the operator through the steps for establishing the system calibration in terms of volts/radian of phase change. The calibration factor can be entered by the operator for either a frequency discrimination set up or a quadrature phase detection set up. If the beat frequency signal from a quad detection set up is a reasonably good sine wave (harmonics less than -30dB) the level can be measured by the analyzer and used as a reference. The zero crossing slope is computed from the measurement and printed out so that the beat frequency signal level need not be measured each time the program is run with the same set up. The subprogram also accounts for any external preamplification which may be necessary to increase measurement sensitivity.

The SET UP subprogram allows the operator to specify the frequency range of interest and the spacing and number of frequency points. Log frequency steps are normally used to cover a wide frequency range with constant percentage resolution at each frequency. Usually 10-20 steps per decade provides adequate resolution for determining the shape of the phase noise spectral density curve. Linear frequency steps are useful for narrow frequency ranges, particularly if harmonic relationships of deterministic signals are being examined.

The MEASURE subprogram computes the frequency points according to the data from the previous inputs, selects an appropriate resolution bandwidth and makes the noise level measurements. Measurements at frequencies below 1000Hz require the 3Hz resolution bandwidth to adequately resolve power line harmonics and other deterministic signals which are often closely spaced at low frequencies. The disadvantage of using the 3Hz bandwidth is that it takes approximately 32 seconds to complete each measurement. Between 1kHz and 10kHz, the 30Hz resolution bandwidth is used to reduce the measurement time to 3 seconds. Above 10kHz where frequency resolution is not as important, the 1kHz bandwidth is used to make measurements in less than 500 milliseconds. Wider bandwidths would not speed the total measurement sequence appreciably due to the I/O time required for frequency programming and measurement data transfer.

The user should be aware that the measured amplitude of deterministic signals will not be accurate using the noise level function because of the bandwidth normalization and detector correction factors used. The correct level can be obtained by applying the following corrections:

| Frequency | Add to Reading |
|----------------------------|----------------|
| 20Hz to 999Hz (3Hz BW) | 2.7 dB |
| 1kHz to 9.999kHz (30Hz BW) | 12.7 dB |
| 10kHz to 40MHz (1kHz BW) | 27.9 dB |

If there are many deterministic signals in a sideband spectral density plot, it might be simpler to turn off the noise level function (delete NL1 in line 570) and make a separate plot which will show the true amplitude of the deterministic signals but not the broadband noise level.

The PLOT routine provides a choice of amplitude scale for the frequency range already determined in the SET UP routine. The log frequency scale is labeled at decade increments whereas the linear scale is labeled at the end points. There is a great deal of flexibility here for the creative programmer to customize the PLOT routine for his particular needs. Using the relationships described earlier, the plot can be made in terms of $S_V(f)$ or $S_Y(f)$ rather than $\mathcal{L}(f)$ as in the example software. Another variation might be a GO/NO GO limit test routine for production testing instead of a plot routine.

Appendix A

This program is suitable for both quadrature detection and frequency discrimination setups. The only basic difference is the system sensitivity calibration factor which must be measured with other equipment. Once that calibration factor is measured, it can be treated as a constant until the setup configuration changes. The baseband frequency measurements are made at spot frequencies with the choice of logarithmic or linear frequency spacing. The plot routine uses the frequency axis covered by the measurement routine but allows the operator to choose the amplitude scale, depending on the need. To simplify operator interaction, the expected alphanumeric responses have been underlined in the "INPUT" statements.

The CALIBRATE routine first takes care of the overhead involved with defining a memory array, putting the analyzer in remote and resetting its controls. The operator then inputs the appropriate constant for the setup being used. This part of the program could be further automated, although more easily for quadrature detection than for frequency

discrimination. For example, the zero crossing slope of the beat signal from the unlocked quadrature phase detector is the same as the peak voltage if the signal is a reasonable good sine wave (harmonic <30 dB).

It is possible to expand the program capability to search for the beat signal, measure the harmonics to check the previous assumption; then, if it is met, measure the signal level and add 3dB to obtain the required constant.

The frequency discriminator case would require measuring the output of the discriminator with intentional sine wave modulation on the carrier at a known frequency and Bessel null. The constant is calculated from the values of the modulation index, the modulating frequency and the measured voltage (see Reference No. 1). With a programmable RF switch, it would also be possible to switch external preamplification in and out to automatically measure its gain. The final output statement sets up the HP 3585A for autoranging to the correct input level while the next routine is being exercised.

```
10  DISP "HP 3585A/9845B PHASE NOISE MEASUREMENT ROUTINE.  PRESS CONT TO GO."
20  PAUSE
30  ! CALIBRATE !
40  INTEGER A(1001) ! Defines data array.
50  An=713 ! Defines analyzer address.
60  REMOTE An
70  LOCAL LOCKOUT ?
80  RESET An ! Preset analyzer controls.
90  INPUT "ARE YOU USING QUAD DETECTION OR DISCRIMINATION ?",A#
100 IF A#[1,1]="Q" THEN Qquad
110 INPUT "WHAT IS THE DISCRIMINATOR CONSTANT [HERTZ/VOLT]?",H
120 GOTO Pre
130 Qquad: INPUT "WHAT IS THE BEAT SIGNAL ZERO CROSSING SLOPE [VOLTS/RAD]?",V
140 R=20*LGT(V)-3 ! Converts entry to log.
150 DISP "SET THE SOURCES TO QUADRATURE, THEN PRESS CONT."
160 PAUSE
170 Pre: INPUT "ARE YOU USING A PREAMP (YES OR NO)?",B#
180 IF B#[1,1]="N" THEN Ipt
190 INPUT "WHAT IS THE GAIN OF THE PREAMP [DB] ?",G
200 Ipt: DISP "CONNECT THE SIGNAL TO THE 1 MEGOHM INPUT, THEN PRESS CONT."
210 OUTPUT An;"I1 RL DV" ! Selects 1 Megohm input,
220 ! & dBv reference level.
230 PAUSE
```

The SETUP routine is intended for fast, flexible, baseband frequency coverage through the use of spot frequency measurements at logarithmic or linear spacings. The disadvantage is that discrete spurious signals may distort the curve drawn through the measurement data or that significant deviations from the interpolated curve may be missed if the spacing is too wide. It is, therefore, advisable during the initial stages of testing to make at least one measurement with a large number of steps to obtain

more detail.

Large, discrete, spurious signals can be observed on the analyzer CRT following the completion of the single sweep initiated in the latter part of this routine. However, because the internal sweep is limited to a linear sweep, the frequency resolution will be quite limited on the low end if a wide frequency range has been chosen. This problem is overcome by using the LOG stepped frequency option of the routine.

```

240      ! SETUP !
250  INPUT "WHAT IS THE START FREQUENCY [HZ]?",F1
260  INPUT "WHAT IS THE STOP FREQUENCY [HZ]?",F2
270  INPUT "DO YOU WANT LOG OR LINEAR FREQUENCY STEPS ?",D#
280  INPUT "HOW MANY STEPS (1000 MAX) ?",S
290  IMAGE "AR0 S2 FB",8D,"HZ FA",8D,"HZ CA S2"      ! Turns input autoranging
300                                                    ! off, sets single sweep,
310                                                    ! loads stop & start freq,
320                                                    ! clears trace & starts
330  OUTPUT An USING 290;F2,F1                      ! single sweep.
340 Sta: OUTPUT An;"D6 T4"                          ! Sets & triggers status
350                                                    ! word output.
360  IF BIT(READBIN(An),7)=1 THEN Sta                ! Checks status bit 7.

```

The actual measurements made during the MEASURE routines are at spot frequencies in the MANUAL sweep and NOISE LEVEL modes of the analyzer. There is no need to correct for analyzer detection characteristics or resolution bandwidth in the routine because that is done internally to the analyzer. The resolution bandwidth selection portion of the routine is intended to provide sufficient resolution for power line related spurious signals at frequencies up 1 kHz then increase the resolution

bandwidth at higher frequencies to improve measurement speed. The frequency breakpoints or resolution bandwidths can be easily modified by the user because the analyzer automatically allows for the proper amount of settling time for an accurate measurement. Although $L(f)$ is the most common expression for phase noise, it is by no means the only one possible for this routine. However, be sure that the calibration factor units correspond with the expression being computed.

```

370      ! MEASURE !
380  DISP "MEASUREMENTS IN PROGRESS."
390  FIXED 0
400  OUTPUT An;"S3 NL1"                              ! Sets manual frequency
410                                                    ! & and noise level modes.
420  B1=0                                              ! Res bandwidth variable.
430  FOR I=0 TO S
440  IF D#[1,2]="LI" THEN Lin
450  F=F1*10^((LGT(F2)-LGT(F1))*I/S)                  ! Calculates log frequency.
460  GOTO Bdw
470 Lin: F=F1+I*(F2-F1)/S                             ! Calculates linear freq.
480 Bdw: B=3                                           ! Selects 3, 30, or 1000
490  IF F<1000 THEN Beb                               ! Hertz res bandwidth de-
500  B=30                                              ! pending on frequency.
510  IF F<1E4 THEN Beb
520  B=1000
530 Beb: IF B=B1 THEN Frq                             ! Skips res bw output if
540                                                    ! no change.
550  B1=B
560  IMAGE "RB",4D,"HZ"
570  OUTPUT An USING 560;B                             ! Outputs new res bw.
580 Frq: IMAGE "S3",8D,"HZ FA HZ D1 T5"
590  OUTPUT An USING 580;F                             ! Outputs manual frequency
600                                                    ! & triggers measurement.
610  ENTER An;X                                       ! Enters measurement data.
620  A(I)=X-G-R-6                                     ! Computes L(f) for quad
630                                                    ! det & stores in array.
640  IF A#[1,1]="Q" THEN Prt
650  A(I)=X-G+20*LGT(H/F)-3                           ! Computes L(f) for freq
660                                                    ! discrim & stores in array.
670 Prt: PRINT F,A(I)
680  NEXT I

```

Although the PLOT routine has more lines than the rest, it can be greatly simplified by removing the labeling and axis portions. For repetitive production measurements, it is much faster to use preprinted plotter paper and output the data directly to a digital plotter. The individual data points are not

connected with a smooth curve intentionally to point out that the data is accurate only for each spot frequency and that assumptions about phase noise levels between the points cannot be made unless more data points have been taken to establish the absence of spurious responses.

```

690      ! PLOT !
700      INPUT "WHAT IS TOP OF VERTICAL SCALE- YMAX [DB]?",Y2
710      INPUT "WHAT IS BOTTOM OF VERTICAL SCALE- YMIN [DB]?",Y1
720      PLOTTER IS "GRAPHICS"
730      GRAPHICS
740      MOVE 12,97                                ! Begin label routine.
750      LABEL "SINGLE SIDEBAND PHASE NOISE      DBC/HZ"
760      LOCATE 10,120,15,95
770      FRAME
780      LDIR PI/2
790      LORG 8
800      IF D#[1,2]="L0" THEN Log
810      SCALE 0,S,Y1,Y2                          ! Draws linear freq axis.
820      GRID S/10,10,0,Y1
830      MOVE 0,Y1
840      LABEL USING "KX";F1
850      MOVE S,Y1
860      LABEL USING "KX";F2
870      GOTO Yax
880      Log: SCALE LGT(F1),LGT(F2),Y1,Y2         ! Draws log freq axis.
890      GRID 1,10,0,Y1
900      FOR I=INT(LGT(F1)) TO INT(LGT(F2))
910      MOVE I,Y1
920      LABEL USING "DEX";10^I
930      NEXT I
940      Yax: LDIR 0                                ! Draws ampl axis.
950      SCALE 0,S,Y1,Y2
960      FOR I=Y1 TO Y2 STEP 10
970      MOVE 0,I
980      LABEL USING "KX";I
990      NEXT I                                    ! End label routine.
1000     Plt: LORG 5                                ! Plot data routine.
1010     LINE TYPE 1
1020     CSIZE 2.3
1030     MOVE 0,A(0)
1040     FOR I=0 TO S
1050     DRAW I,A(I)
1060     LABEL USING "K";"X"
1070     MOVE I,A(I)
1080     NEXT I
1090     LOCAL An                                  ! Sets analyzer to local.
1100     END

```

The program included for the HP 85 Personal Computer is essentially the same as the HP 9845T program except that it has been abbreviated slightly, particularly in the INPUT statements and the PLOT routine. Due to the limited resolution of the CRT, the grid lines have been deleted and the axes are drawn right to the limits of the screen. Consequently, the axes labeling was dropped because of the

lack of space. Also, since it is more difficult to precisely locate the center of a label using the HP 85, the point plotting was replaced by a continuous line drawn between the data points. Remember, however, that this does not mean that the line necessarily represents the level of phase noise which actually exists between the points. More points should be used to verify the actual level.

```

10 DISP "      * PHASE NOISE ROUT
   INE *          (PRESS CONT T
   O START)"
20 PAUSE
30 ! "CAL"
40 INTEGER A(256)
50 A1=713
60 REMOTE A1
70 CLEAR 7
80 DISP "QUAD DET OR DISCRIM";
90 INPUT A#
100 IF A#[1,13]="0" THEN 140
110 DISP "DISCRIM CONSTANT (V/HZ
   )";
120 INPUT H
130 GOTO 190
140 DISP "BEAT SIGNAL SLOPE (V/R
   AD)";
150 INPUT V
160 R=20*LG(T(V))-3
170 DISP "SET QUAD, THEN CONT."
180 PAUSE
190 G=0
200 DISP "PREAMP";
210 INPUT B#
220 IF B#[1,13]="N" THEN 250
230 DISP "GAIN (DB)";
240 INPUT G
250 DISP "CONNECT SIGNAL TO 1 MO
   HM INPUT, THEN CONT."
260 OUTPUT A1 ; "I1 RL DV"
270 PAUSE
280 ! SETUP !
290 DISP "START FREQUENCY (HZ)";
300 INPUT F1
310 DISP "STOP FREQUENCY (HZ)";
320 INPUT F2
330 DISP "LOG OR LINEAR STEPS";
340 INPUT D#
350 DISP "NUMBER OF STEPS";
360 INPUT S
370 IMAGE "AR0 S2 FB",80,"HZ FA"
   ,80,"HZ CA S2"
380 OUTPUT A1 USING 370 ; F2,F1
390 OUTPUT A1 ; "DG T4"
400 IMAGE #,B
410 ENTER A1 USING 400 ; S1
420 IF BIT(S1,7)=1 THEN 390
430 ! MEASURE !
440 DISP "MEASUREMENTS IN PROGRE
   SS"

```

```

450 OUTPUT A1 ; "S3 NL1"
460 B1=0
470 FOR I=0 TO S
480 IF D#[1,2]="LI" THEN 510
490 F=F1*10^((LGT(F2)-LGT(F1))*I
   /S)
500 GOTO 520
510 F=F1+I*(F2-F1)/S
520 B=3
530 IF F<1000 THEN 580
540 B=30
550 IF F<10000 THEN 580
560 B=1000
570 IF B=B1 THEN 610
580 B1=B
590 IMAGE "RB",40,"HZ"
600 OUTPUT A1 USING 590 ; B
610 IMAGE "S3",80,"HZ FA HZ D1 T
   5"
620 OUTPUT A1 USING 610 ; F
630 ENTER A1 ; X
635 IF A#[1,13]="0" THEN 660
640 A(I)=X-G-R-6
650 GOTO 670
660 A(I)=X-G+20*LG(T(H/F))-3
670 PRINT USING "3DC3DC3D,7X,4D"
   ; F,A(I)
680 NEXT I
690 PRINT USING "4/"
700 ! PLOT !
710 DISP "YMAX (DB)"
720 INPUT Y2
730 DISP "YMIN (DB)"
740 INPUT Y1
750 GCLEAR
760 SCALE 0,100,0,100
770 MOVE 20,90
780 LABEL "PHASE NOISE DBC/HZ"
790 IF D#[1,2]="LO" THEN 840
800 SCALE 0,100,Y1,Y2
810 XAXIS Y1,10
820 YAXIS 0,10
830 GOTO 870
840 SCALE LGT(F1),LGT(F2),Y1,Y2
850 XAXIS Y1,1
860 YAXIS LGT(F1),10
870 SCALE 0,8,Y1,Y2
880 FOR I=0 TO S
890 PLOT I,A(I)
900 NEXT I
910 END

```

Appendix B

Accounting for Analyzer Characteristics

There are a variety of methods for measuring phase noise in the frequency domain but the common tool used in all of them is a frequency selective analyzer. It is appropriate now to discuss the different types of analyzers and the corrections necessary for making accurate noise measurements.

Wave analyzers are generally manually tuned selective analyzers with meter readout and flat-top, steep-sided IF filters. The tuning range on the low end is limited to about 5 times the narrowest IF filter available in the instrument. Since the IF filters are not ideally rectangular, it is necessary to know the equivalent noise bandwidth in order to normalize to a 1Hz bandwidth. For wave analyzers, the equivalent noise bandwidths are typically 3 to 10% wider than the stated 3dB bandwidth. Since this can vary from unit to unit, it is wise to check the actual unit by numerical integration of the filter curve to at least 30dB down from the top of the filter.

Nearly all analyzers use an average detector calibrated to read the true rms level of a discrete signal in the passband. For white noise, however, the meter reading would be 1.05dB lower than the true level. Since the meter may be fluctuating randomly, it is necessary to visually average the reading if some form of video filtering or meter damping is not available. Wave analyzers are most useful where only a relatively few spot measurements are necessary to verify a phase noise spectral density.

Spectrum analyzers are automatically swept selective analyzers with a CRT display. The IF filters are gaussian shaped for fast settling and rapid sweeping. Since the skirts are wider than wave analyzers, the equivalent noise bandwidths are usually up to 15% more than the 3dB bandwidths. Spectrum analyzers have logarithmic IF amplifier gain which amplifies noise peaks less than lower values and produces a signal which when average detected requires a total of +2.5dB correction for white noise. Video filtering, which is available on most spectrum analyzers is useful for reducing the amplitude deviation of the spectral density.

Analyzers which are remotely programmable and have digital data output can be used for

numerically averaging many readings at a single frequency with software in the controller. It is necessary, of course, to retain statistical independence of the samples by limiting the repetition rate of the readings to the reciprocal of the equivalent bandwidth of the IF filter bandwidth and the video filtering bandwidth combined. From statistical theory, the confidence in an average is improved by the square root of the number of samples. For example, the average of 100 samples is 10 times better than a single sample.

Amplitude sensitivity is sometimes a problem when measuring very low phase noise levels. However, there are readily available low noise preamps to provide the necessary gain. This additional gain must be removed during calibration but it is a small inconvenience.

To summarize, for wave analyzers, the corrected noise power reading in dB is:

$$\text{Noise} = \text{Meter Reading} + 1.05 - 10 \log (\text{Equiv. Noise BW}) - \text{Preamp Gain}$$

For spectrum analyzers, it is:

$$\text{Noise} = \text{CRT display} + 2.5 - 10 \log (\text{Equiv. Noise BW}) - \text{Preamp Gain}$$

Remember, however, that all these corrections apply only for signals which approximate white noise in the IF bandwidth being used. Deterministic components are discrete signals which do not require correction factors. Whenever a data point is above the adjacent points by several dB, it should be checked for discreteness by narrowing the IF bandwidth and widening the video filtering. A discrete signal will not change level when the IF bandwidth is narrowed if it is in the center of the passband and will fluctuate less than adjacent points when the amount of video filtering is reduced.

The HP 3585A Spectrum Analyzer also requires the above correction for noise measurements in the standard spectrum analysis mode. However, the noise level mode used with the marker readout automatically includes the correction and displays the amplitude normalized to a 1Hz bandwidth.

References

1. "Wave Analyzer Dynamic Range and Bandwidth Requirements for Signal Noise Analysis", Technical Report RE 76-26, U.S. Army Missile Command, Redstone Arsenal, Alabama 35809.
2. "Frequency Domain Stability Measurements: A Tutorial Introduction", NBS Technical Note 679, U.S. Department of Commerce/National Bureau of Standards.
3. "Frequency Stability Specification and Measurement: High Frequency and Microwave Signals", NBS Technical Note 632, U.S. Department of Commerce/National Bureau of Standards.
4. "Characterization of Frequency Stability", NBS Technical Note 394, U.S. Department of Commerce/National Bureau of Standards.
5. "Spectrum Analysis...Noise Measurements", Spectrum Analyzer Series Application Note 150-4, Hewlett-Packard Company, April, 1974.
6. "Short-Term Stability for a Doppler Radar: Requirements, Measurements, and Techniques", *PROCEEDINGS OF THE IEEE* Vol. 54, No. 2, February, 1966. D.B. Leeson, Senior Member, IEEE, and G.F. Johnson, Senior Member, IEEE.



AN 246-2

For more information, call your local HP Sales Office or nearest Regional Office: • Eastern (201) 265-5000; • Midwestern (312) 255-9800; • Southern (404) 955-1500; • Western (213) 970-7500; • Canadian (416) 678-9430. Ask the operator for instrument sales. Or write Hewlett-Packard, 1501 Page Mill Road, Palo Alto, CA 94304. **In Europe:** Hewlett-Packard S.A., 7, rue du Bois-du-Lan, P.O. Box, CH 1217 Meyrin 2, Geneva, Switzerland. **In Japan:** Yokogawa-Hewlett-Packard Ltd., 29-21, Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.